Applied Psychological Measurement
36(3) 237–248

DOI: 10.1177/0146621612441857
http://apm.sagepub.com

Using the GLIMMIX Procedure in SAS 9.3 to Fit a Standard Dichotomous Rasch and Hierarchical 1-PL IRT Model

Ryan A. Black[1] and Stephen F. Butler[1]

Abstract

Although Rasch models have been shown to be a sound methodological approach to develop and validate measures of psychological constructs for more than 50 years, they remain underutilized in psychology and other social sciences. Until recently, one reason for this underutilization was the lack of syntactically simple procedures to fit Rasch and item response theory (IRT) models in general statistical software packages. In this article, the authors demonstrate how to fit the standard dichotomous Rasch model and a dichotomous one-parameter logistic IRT model with nested random effects via the easy-to-use GLIMMIX procedure in SAS 9.3. For comparison purposes, the standard dichotomous Rasch model was also fit using the Rasch specialized software, WINSTEPS 3.68.2. The SAS code used to simulate the data on which the Rasch model was fit is provided to allow replication of estimates. Findings suggest that the GLIMMIX procedure may be a viable option for fitting the standard dichotomous Rasch and dichotomous IRT models.

Keywords

standard dichotomous Rasch model, dichotomous 1-PL IRT model with nested random effects, item response theory, IRT, SAS 9.3, GLIMMIX

In this article, the authors demonstrate how to fit a standard dichotomous Rasch model, using the GLIMMIX procedure in SAS 9.3. For comparison purposes, the same model was refitted using the Rasch specialized software, WINSTEPS 3.68.2 (Linacre, 2009). Although the focus of this article is on the standard Rasch model, the authors also used a dichotomous one-parameter logistic (1-PL) item response theory (IRT) model with nested random effects to demonstrate some of GLIMMIX's additional capabilities of IRT modeling.

[1]Inflexxion, Inc., Newton, MA, USA

Corresponding Author:
Ryan A. Black, Inflexxion, Inc., 320 Needham Street, Suite 100, Newton, MA 02464, USA
Email: rblack@inflexxion.com

Method

Standard Dichotomous Rasch Model

In the standard dichotomous Rasch equation, the probability of a positive endorsement is a function of person ability (''theta'') minus item difficulty (''b'') (Bond & Fox, 2007; Rasch, 1960):

$$\text{logit}_{ij} = \log\left[\text{p}_{ij}/\left(1 - \text{p}_{ij}\right)\right] = \theta_j - \text{b}_i, \tag{1}$$

where

θ_j = fixed effect of the jth person and

b_i = fixed effect of the ith item.

The marginal likelihood of the standard dichotomous Rasch model was approximated using the maximum likelihood estimation (MLE) offered in the GLIMMIX procedure (Schabenberger, 2007; Figure 1). Variances and residuals on the probability scale were outputted from GLIMMIX to calculate person and item infit and outfit mean-square statistics. These fit statistics were computed via the SQL procedure (Figure 2) based on equations provided in the WINSTEPS documentation (Linacre, 2009). An item map is also provided to illustrate the hierarchy of the items along the continuum of the measure (SAS code in Figure 3a), whereas a histogram is provided to show the distribution of person abilities (SAS code in Figure 3b). The standard dichotomous Rasch model was also fit on the same data in WINSTEPS 3.68.2 using joint maximum likelihood estimation (JMLE) for comparison purposes.

Data Simulation for Standard Dichotomous Rasch Model

Data were simulated in SAS to approximate a standard dichotomous Rasch model by applying the following specifications: (a) 1,000 persons, (b) responses to 50 dichotomous items from each person, (c) normally distributed person abilities with means of 0 logits and standard deviations of 1.25 logits, (d) item difficulties ranging from a probability of .01 (Item 1—very difficult to endorse) to a probability of .99 (Item 50—very easy to endorse), (e) the log odds of endorsing an item as a function of person abilities (''theta'') minus item difficulty (''b''), and (f) discrimination parameters fixed at 1.0 (Figure 4).

Dichotomous 1-PL IRT Model With Nested Random Effects

In a dichotomous 1-PL IRT model with persons *nested* within a 3rd-level unit (e.g., geographic location), the probability of endorsement is a function of item, person, and 3rd-level unit parameters (e.g., geographic location of person). Notably, item effects were treated as fixed effects, whereas person and location effects were treated as random:

$$\text{logit}_{ijk} = \log\left[\text{p}_{ijk}/\left(1 - \text{p}_{ijk}\right)\right] = \theta_{j(k)} + \text{l}_k - \text{b}_i, \tag{2}$$

where

$\theta_{j(k)}$ = random effect of the jth person in the kth location, $\text{Person}_{j(k)} \sim i.i.d.N\left(0, \sigma^2{}_{\text{Person}}\right)$;

```
/*Glimmix Code*/

proc glimmix data = Rasch /*method = laplace*/;
    class item person /*location*/;
    model response(descending) = item person / s noint link=logit dist=binary;
    lsmeans item / cl ilink;
    lsmeans person / cl ilink;
    output out=fit_statistics resid(ilink)=residual variance(ilink)=variance;
    /*random int / subject = person(location) s;*/
    /*random int / subject = location s;*/
run;
```

Figure 1. SAS code to fit standard dichotomous Rasch model in GLIMMIX
1. The LSMEANS statements provide the item abilities and person difficulties in logits and probability units.
2. To conform to the standard Rasch model in which item effects are interpreted as item difficulties, it is necessary to reverse the direction of the item effects from the LSMEANS statement. Although not provided, this can be accomplished easily using a SAS Data Step.
3. This code fits a standard dichotomous Rasch model. To incorporate nested random effects for the dichotomous one-parameter logistic (1-PL) model discussed in this article, one would need to remove "/* . . . */" as well as "person" from the MODEL statement.

l_k = random effect of the kth location, $\text{location}_k \sim i.i.d. N(0, \sigma^2_{\text{Location}})$; and

b_i = fixed effect of the ith item.

The marginal likelihood of this model was approximated using a frequentist approach by employing the Laplace method offered in GLIMMIX (Schabenberger, 2007).[1] The GLIMMIX code was specified to estimate the item difficulties and person abilities in probability units and logits. Given the focus of this article on the standard dichotomous Rasch model, a limited amount of information regarding fit is provided with respect to this model.

Data Simulation for Dichotomous 1-PL IRT Model With Nested Random Effects

Data were simulated in SAS to approximate a dichotomous 1-PL IRT model with nested random effects by applying the following specifications: (a) 250 randomly selected persons *nested* in each of 50 randomly selected locations ($N_{\text{person}} = 250 \times 50 = 12{,}500$), (b) responses to 50 dichotomous items from each person, (c) normally distributed location ability (location random effect) and person ability (person random effect) scores with means of 0 logits and standard deviations of 1.25 logits, (d) item difficulties ranging from a probability of .01 (Item 1—very difficult to endorse) to a probability of .99 (Item 50—very easy to endorse), and (e) the log odds of endorsing an item as a function of location and person abilities minus item difficulty. Data simulation code for this model is available on request.

Handling of Missing Data in GLIMMIX

In the data set, the response variable (1 = *endorse*, 0 = *not endorse*) is concatenated vertically and linked to both person and item indicator variables, such that each participant identification number repeats for each item. As a result, if a participant has responded to 45 of the 50 items, for instance, there should be 45 cases with valid response values and 5 cases with missing response values. The 45 cases with valid response data would be used in the parameter

```
/*Item and Person Infit and Outfit Mean-Square Code*/

data fit_statistics;
 set fit_statistics;
 standardized_residual = residual / sqrt(variance);
run;

/*Estimated Item Mean Squares*/
proc sql;
 create table fit_statistics as
 select person, item, response, residual, variance, standardized_residual,
      sum(standardized_residual**2)/1000 as item_outfit_msqr
 from fit_statistics
 group by item;
quit;
proc sql;
 create table fit_statistics as
 select person, item, response, residual, variance, standardized_residual, item_outfit_msqr,
       sum(variance*(standardized_residual)**2)/sum(variance) as item_infit_msqr
 from fit_statistics
 group by item;
quit;

/*Estimated Person Mean Squares*/
proc sql;
 create table fit_statistics as
 select person, item, response, residual, variance, standardized_residual, item_outfit_msqr,
item_infit_msqr,
       sum(standardized_residual**2)/50 as person_outfit_msqr
 from fit_statistics
 group by person;
quit;
proc sql;
 create table fit_statistics as
 select person, item, response, residual, variance, standardized_residual, item_outfit_msqr,
item_infit_msqr, person_outfit_msqr,
      sum(variance*(standardized_residual)**2)/sum(variance) as person_infit_msqr
 from fit_statistics
 group by person;
quit;
```

Figure 2. SAS code to produce item and person infit and outfit mean-squares
Note: The code above uses the SQL procedure in SAS to replicate the equations provided in the *WINSTEPS User Guide* for constructing item and person infit and outfit mean-square statistics (Linacre, 2009).

estimation from GLIMMIX. That is, all available valid data from a given participant can be used when using the GLIMMIX procedure.

Results

Standard Dichotomous Rasch Model Fit in SAS Using the GLIMMIX Procedure

Convergence and overall fit of models. The standard dichotomous Rasch model using MLE in GLIMMIX converged in 6 min. The $-2 \times$ Log Likelihood ($-2 \times$ LL) for this model was 37,781. This analysis was run on a 64-bit Windows 7.0 operating system with 8 gigabytes of RAM.

```
/*Item Map*/
proc gplot data=lsmeans_item_effects;
symbol1 v= star i=join;

Title "Item Map";
axis1 label = (angle=90 "Logit");
axis2 order = (1 to 50 by 1);

plot estimate*item/vaxis=axis1 haxis=axis2;
format estimate item;
run;
```

Figure 3a. SAS code to generate item map
Note: For this code to run, LSMEANS data sets need to be outputted from GLIMMIX using ODS statements.

Item-level statistics: Difficulties, standard errors, infit and outfit mean-squares, and item map. Estimated non-centered and centered* item difficulties, standard errors, and item infit and outfit mean-squares are presented in Table 1. Although not shown, 95% confidence intervals for the item difficulty estimates from this model included the true parameter for all 50 items. Also, none of the item infit or outfit statistics produced by GLIMMIX is below 2.0 (Linacre, 2009). The item map (Figure 5a) shows coverage along the entire continuum as specified in the simulation code.

Person-level statistics: Standard errors, infit and outfit mean-squares, and abilities. In lieu of presenting statistics in a Table for all 1,000 participants, the following is noted: (a) 100% of the infit mean-squares and 94.5% of the outfit mean-squares are below 2.0 and (b) the distribution of person abilities is approximately normal (Figure 5b).[2]

Standard Rasch Model Fit in WINSTEPS

Convergence and overall fit of model. The standard Rasch analysis performed in WINSTEPS 3.68.2 converged in less than 1 min, yielding a fit of $-2 \times$ LL = 37,781. Notably, the $-2 \times$ LL generated from WINSTEPS is identical (when rounded to the whole number) to the $-2 \times$ LL from GLIMMIX.

Item-level statistics: Item difficulties, standard errors, infit and outfit mean-squares, and item map. Estimated item difficulties, standard errors, and item infit and outfit mean-squares are presented in Table 1. As seen in Table 1, the estimates obtained from the standard Rasch model used in WINSTEPS are similar to those obtained from the standard models fit in SAS. 95% confidence intervals for the item difficulty estimates from this model also included the true parameter for all 50 items. Similar to the GLIMMIX results, none of the item infit or outfit statistics reaches the threshold of 2.0. Although not presented here, the item map from WINSTEPS reveals essentially the same hierarchy as the item map produced from SAS.

Person-level statistics: Abilities, infit and outfit mean-squares, and distribution. Similar to GLIMMIX, 100% of the infit and 94.5% of outfit mean-squares were below 2.0, and the distribution of person abilities was approximately normal (not shown). It is also noted that the

*For comparison purposes, item difficulties were centered at 0 as this is the default for Winsteps (Linacre, 2009)

```
/*Person Ability Distribution Histogram*/
proc univariate data= lsmeans_person_effects;

Title "Person Ability Distribution";
footnote "Person Ability";
axis1 label = (angle=90 "Logit");

histogram estimate / normal;
run;


Note: In order for this code to run, LSMEANS datasets need to be outputted from GLIMMIX
employing ODS statements.
```

Figure 3b. SAS code to generate person abilities histogram
Note: For this code to run, LSMEANS data sets need to be outputted from GLIMMIX using ODS statements.

```
/*SAS Simulation Code*/

data Rasch;
  array item__ {50} item1-item50;
  array item_prob__ {25} _temporary_
      (0.010  0.015  0.020  0.025  0.030  0.035  0.050  0.070  0.080  0.100
       0.125  0.150  0.175  0.200  0.225  0.250  0.275  0.300  0.325  0.350
       0.375  0.400  0.425  0.450  0.475);
  array item_difficulty__ {50} item_difficulty1-item_difficulty50;
  do i=1 to 25;
    item_difficulty__{i} = log((item_prob__{i}) / (1 - item_prob__{i}));
    item_difficulty__{51-i} = -item_difficulty__{i};
  end;

      do person=1 to 1000;
      person_ability = 1.25*rannor(48792371);
      do item=1 to 50;
        do j=1 to 50;
          item__{j} = (item=j);
        end;

        eta = person_ability - item_difficulty__{item};
        prob = exp(eta)/(1+exp(eta));
        response = ranbin(48792371,1,prob);
        output;
      end;
    end;
 keep response person item;
run;
```

Figure 4. SAS simulation code



Figure 5a. Item map generated from SAS



Figure 5b. Person abilities histogram generated from SAS

Pearson correlation between the person abilities and infit and outfit mean-squares produced from WINSTEPS and GLIMMIX were approximately 1.00, 1.00, and 0.98, respectively.

1-PL IRT With Nested Random Effects in GLIMMIX

Convergence and overall fit of model. The 1-PL IRT model with nested random effects using the Laplace method in GLIMMIX converged after 2 hr and 22 min. This model yielded an overall fit statistic of $-2 \times \text{LL} = 482{,}539$.

Item difficulties and person abilities. As seen in Table 2, the estimated item difficulties from the GLIMMIX procedure were similar to the simulation parameters. 95% confidence intervals for the item difficulty estimates from this model included the true parameter for all 50 items. Although not shown, it is noted that person abilities approximated a normal distribution similar to the specifications from the simulation.

Discussion

In this article, the authors demonstrated how to fit a standard dichotomous Rasch model via the GLIMMIX procedure in SAS 9.3 and compared the overall fit and item- and

Table 1. Item Difficulties (in Logits), Standard Errors, and Infit and Outfit Mean-Squares Estimated From Standard Dichotomous Rasch Models Fitted in SAS-GLIMMIX and WINSTEPS

Item	SAS simulation parameter	Estimate		*SE*		Infit mean-square		Outfit mean-square	
		GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS
1	−4.60	−4.87 (−4.90)	−4.89	0.25	0.25	1.00	1.00	1.28	1.27
2	−4.18	−4.00 (−4.03)	−4.02	0.17	0.17	1.02	1.01	1.59	1.58
3	−3.89	−3.86 (−3.89)	−3.88	0.17	0.16	1.03	1.03	1.09	1.09
4	−3.66	−3.71 (−3.74)	−3.73	0.16	0.15	1.03	1.03	0.90	0.89
5	−3.48	−3.57 (−3.60)	−3.59	0.15	0.15	0.99	0.98	0.91	0.90
6	−3.32	−3.49 (−3.51)	−3.51	0.14	0.14	0.96	0.96	0.69	0.69
7	−2.94	−3.05 (−3.07)	−3.07	0.12	0.12	1.00	0.99	0.86	0.86
8	−2.59	−2.62 (−2.65)	−2.64	0.11	0.11	0.98	0.98	0.93	0.93
9	−2.44	−2.39 (−2.42)	−2.41	0.10	0.10	0.98	0.97	1.00	1.00
10	−2.20	−2.21 (−2.24)	−2.23	0.10	0.10	0.96	0.96	0.85	0.84
11	−1.95	−1.98 (−2.00)	−2.00	0.09	0.09	1.03	1.03	0.99	0.99
12	−1.73	−1.61 (−1.64)	−1.63	0.09	0.09	1.00	1.00	1.04	1.04
13	−1.55	−1.62 (−1.65)	−1.65	0.09	0.09	0.98	0.98	0.95	0.95
14	−1.39	−1.29 (−1.32)	−1.32	0.08	0.08	0.98	0.98	0.93	0.93
15	−1.24	−1.31 (−1.33)	−1.33	0.08	0.08	0.97	0.96	0.91	0.91
16	−1.10	−1.13 (−1.16)	−1.16	0.08	0.08	0.95	0.95	0.93	0.93
17	−0.97	−0.91 (−0.94)	−0.94	0.08	0.08	0.97	0.97	0.93	0.93
18	−0.85	−0.81 (−0.83)	−0.83	0.08	0.08	1.00	1.00	1.03	1.03
19	−0.73	−0.74 (−0.77)	−0.77	0.08	0.08	1.00	1.00	1.02	1.01
20	−0.62	−0.53 (−0.56)	−0.56	0.07	0.07	1.00	1.00	0.97	0.97
21	−0.51	−0.38 (−0.41)	−0.41	0.07	0.07	1.00	1.00	1.01	1.01
22	−0.41	−0.50 (−0.52)	−0.52	0.07	0.07	1.00	1.00	0.94	0.94
23	−0.30	−0.30 (−0.33)	−0.32	0.07	0.07	1.08	1.07	1.03	1.03
24	−0.20	−0.26 (−0.29)	−0.29	0.07	0.07	1.01	1.01	1.01	1.01
25	−0.10	−0.14 (−0.16)	−0.16	0.07	0.07	0.95	0.95	0.95	0.95
26	0.10	0.15 (0.12)	0.12	0.07	0.07	1.01	1.01	1.01	1.01
27	0.20	0.22 (0.19)	0.19	0.07	0.07	1.01	1.01	1.02	1.02
28	0.30	0.31 (0.29)	0.29	0.07	0.07	1.00	1.00	1.05	1.05
29	0.41	0.53 (0.51)	0.51	0.08	0.07	0.98	0.98	1.06	1.06

(continued)

Table 1. (continued)

Item	SAS simulation parameter	Estimate		*SE*		Infit mean-square		Outfit mean-square	
		GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS	GLIMMIX	WINSTEPS
30	0.51	0.53 (0.51)	0.51	0.08	0.07	1.03	1.03	1.02	1.02
31	0.62	0.54 (0.51)	0.51	0.08	0.07	1.02	1.01	1.10	1.10
32	0.73	0.73 (0.70)	0.70	0.08	0.08	1.06	1.06	1.02	1.02
33	0.85	1.00 (0.97)	0.97	0.08	0.08	1.03	1.03	1.17	1.16
34	0.97	0.96 (0.93)	0.93	0.08	0.08	0.97	0.97	0.98	0.98
35	1.10	1.06 (1.03)	1.03	0.08	0.08	1.01	1.01	1.00	1.00
36	1.24	1.30 (1.27)	1.27	0.08	0.08	0.99	0.99	0.92	0.92
37	1.39	1.33 (1.30)	1.30	0.08	0.08	1.00	1.00	0.95	0.95
38	1.55	1.53 (1.51)	1.50	0.08	0.08	0.96	0.95	0.86	0.86
39	1.73	1.84 (1.81)	1.81	0.09	0.09	1.04	1.04	1.10	1.09
40	1.95	1.83 (1.80)	1.80	0.09	0.09	1.06	1.06	1.04	1.04
41	2.20	2.30 (2.28)	2.27	0.10	0.10	0.98	0.98	1.19	1.18
42	2.44	2.45 (2.42)	2.42	0.10	0.10	0.99	0.98	1.07	1.07
43	2.59	2.51 (2.49)	2.48	0.11	0.10	0.97	0.97	1.01	1.00
44	2.94	3.08 (3.05)	3.05	0.13	0.12	0.96	0.95	0.79	0.79
45	3.32	3.46 (3.44)	3.43	0.14	0.14	1.01	1.00	0.96	0.96
46	3.48	3.48 (3.46)	3.45	0.14	0.14	1.01	1.01	1.15	1.14
47	3.66	3.88 (3.85)	3.84	0.17	0.16	1.00	1.00	1.07	1.06
48	3.89	4.22 (4.19)	4.18	0.19	0.19	0.98	0.98	1.37	1.36
49	4.18	4.41 (4.38)	4.37	0.20	0.20	0.99	0.99	0.96	0.95
50	4.60	4.98 (4.96)	4.95	0.26	0.26	1.03	1.03	1.31	1.30

The estimated item difficulties in parentheses are centered at 0.

Table 2. Item Difficulties (in Logits) Estimated From the Dichotomous 1-PL IRT Model With Nested Random Effects

Item	SAS simulation parameter	GLIMMIX			
		Estimate	*SE*	95% CI	
1	−4.60	−4.63	0.18	−4.99	−4.27
2	−4.18	−4.15	0.18	−4.50	−3.80
3	−3.89	−3.86	0.18	−4.21	−3.50
4	−3.66	−3.71	0.18	−4.06	−3.36
5	−3.48	−3.46	0.18	−3.81	−3.11
6	−3.32	−3.37	0.18	−3.72	−3.02
7	−2.94	−2.95	0.18	−3.30	−2.60
8	−2.59	−2.61	0.18	−2.96	−2.26
9	−2.44	−2.49	0.18	−2.84	−2.14
10	−2.20	−2.21	0.18	−2.55	−1.86
11	−1.95	−1.95	0.18	−2.29	−1.60
12	−1.73	−1.74	0.18	−2.09	−1.40
13	−1.55	−1.55	0.18	−1.90	−1.21
14	−1.39	−1.40	0.18	−1.74	−1.05
15	−1.24	−1.25	0.18	−1.59	−0.90
16	−1.10	−1.06	0.18	−1.41	−0.72
17	−0.97	−0.97	0.18	−1.32	−0.63
18	−0.85	−0.88	0.18	−1.23	−0.54
19	−0.73	−0.74	0.18	−1.08	−0.39
20	−0.62	−0.64	0.18	−0.98	−0.29
21	−0.51	−0.51	0.18	−0.86	−0.17
22	−0.41	−0.44	0.18	−0.78	−0.10
23	−0.30	−0.29	0.18	−0.63	0.06
24	−0.20	−0.24	0.18	−0.58	0.11
25	−0.10	−0.15	0.18	−0.49	0.20
26	0.10	0.06	0.18	−0.28	0.41
27	0.20	0.19	0.18	−0.15	0.54
28	0.30	0.32	0.18	−0.02	0.67
29	0.41	0.42	0.18	0.07	0.76
30	0.51	0.51	0.18	0.16	0.85
31	0.62	0.60	0.18	0.26	0.95
32	0.73	0.74	0.18	0.40	1.09
33	0.85	0.87	0.18	0.53	1.22
34	0.97	0.99	0.18	0.64	1.33
35	1.10	1.09	0.18	0.75	1.44
36	1.24	1.22	0.18	0.88	1.57
37	1.39	1.38	0.18	1.04	1.73
38	1.55	1.54	0.18	1.20	1.89
39	1.73	1.75	0.18	1.40	2.09
40	1.95	1.94	0.18	1.59	2.28
41	2.20	2.15	0.18	1.81	2.50
42	2.44	2.43	0.18	2.08	2.77
43	2.59	2.61	0.18	2.26	2.96
44	2.94	2.92	0.18	2.57	3.27
45	3.32	3.30	0.18	2.95	3.65
46	3.48	3.47	0.18	3.12	3.82
47	3.66	3.70	0.18	3.35	4.05
48	3.89	3.92	0.18	3.57	4.27
49	4.18	4.18	0.18	3.82	4.53
50	4.60	4.64	0.18	4.28	5.00

Note: 1-PL IRT = one-parameter logistic item response theory; CI = confidence interval person-specific estimates with WINSTEPS. Findings suggest that the GLIMMIX procedure produces estimates that are comparable with WINSTEPS. The authors also used a 1-PL IRT model, which incorporated nested random effects (items nested in persons and persons nested in locations).[3] Although not shown, other similar types of models may be fitted in GLIMMIX such as a model with crossed random effects, and a model that treats item effects as a random sample of a population of items (item random effects) may be employed using the GLIMMIX procedure.

A critical assumption of the Rasch model is unidimensionality, which is commonly assessed by performing an unrotated principal components analysis (PCA) on probability scale residuals obtained from the Rasch model (Linacre, 2006). It is possible to perform an unrotated PCA on probability scale residuals (outputted from the GLIMMIX procedure) using the FACTOR procedure in SAS.

Because this was a simulation experiment, the steps to defining a construct were not discussed. However, it should be noted that defining a construct entails examination of (a) item content, coverage, fit, and hierarchy, (b) person fit (including examination of any atypical patterns of responses) and hierarchy, and finally (c) overall model fit and dimensionality.

In conclusion, it is hoped that this demonstration of how to fit a standard dichotomous Rasch model as well as a hierarchical 1-PL model (e.g., nested random effects) in a generalized linear mixed model procedure using a general software package such as SAS will help the field of measurement move forward.

Acknowledgments

The authors wish to thank Dale F. McLerran, MS, for reviewing the initial draft of the manuscript, providing assistance with the development of the SAS code, and statistical consultation.

Declaration of Conflicting Interests

The author(s) declared no potential conflicts of interest with respect to the research, authorship, and/or publication of this article.

Funding

The author(s) received no financial support for the research, authorship, and/or publication of this article.

Notes

1. Although the adaptive quadrature method that allows for multiple quadrature points could yield less biased estimates for generalized linear mixed models than the Laplace method (one quadrature point), given the size of the nested random effects design matrix, the adaptive quadrature method was deemed too computationally intensive.
2. To determine the hierarchy of each individual along the continuum of the measure, please refer to the output produced from the ''LSMEANS person'' statement in the GLIMMIX code from Figure 1.
3. SAS code used to generate the data on which the standard dichotomous Rasch model was fit was provided to allow those with access to SAS 9.3 to replicate the analyses on the same data. SAS simulation code to generate data for the nested random effects model is available on request.

References

Bond, T. G., & Fox, C. M. (2007). *Applying the Rasch model: Fundamental measurement in the human sciences* (2nd ed.). London, England: Erlbaum.

Linacre, J. M. (2006). Data variance explained by Rasch measures. *Rasch Measurement Transactions, 20*(1), 1045.

Linacre, J. M. (2009). WINSTEPS Rasch measurement computer program (Version 3.68.2) [Computer software]. Chicago, IL: Winsteps.com.

Rasch, G. (1960). *Probabilistic models for some intelligence and achievement tests*. Copenhagen, Denmark: Danish Institute.

Schabenberger, O. (2007). Growing Up Fast: SAS® 9.2 Enhancements to the GLIMMIX procedure. In *Proceedings of the 2007 SAS Global Forum,* 1-20.